
Abstract: This document presents an average macro model for the fly-buck converter. The model can be used for both
large and small signal modeling. Parasitic and lossy components are included in the model, and it is partially
based on a conventional average switch model for a buck stage. For isolated output, the analytic solution of
the average current in a secondary winding is proposed. The presented model is implemented in SPICE, and
simulation results are compared to switching model simulation and experimental data.

1 INTRODUCTION

The fly-buck converter has become popular because
it has several advantages, such as good cross regula-
tion, line transient response, and low EMI, (Fang and
Meng, 2015; Karlsson and Persson, 2017; Gu and
Kshirsagar, 2017; Choudhary, 2015; Nowakowski,
2012). It has a simple design and provides multiple
isolated outputs. A small-signal analytical model for
an ideal fly-buck converter was presented in (Wang
et al., 2017), but the effects of component parasitics
could not be predicted.

The proposed model can be used for both large
and small signal analysis and can be simulated in time
or frequency domains. The difficulty of developing
such a model is that leakage inductance current has a
pulsed shape and cannot be approximated with con-
ventional small ripple approximation, (Erickson and
Maksimovic, 2007). To overcome this issue, the cur-
rent is calculated during the instantaneous switching
period, and small ripple approximation is used for the
transformer’s magnetizing inductance current and ca-
pacitor voltages. The model accounts for the losses
and parasitics of semiconductors and magnetics and
has been implemented as a SPICE subcircuit. The fol-
lowing assumptions were considered: the model cov-
ers two isolated outputs, and the dead-time effect is
negligible.

2 MODEL DERIVATION

The fly-buck converter’s basic structure is shown
in Fig. 1. The MOSFETS Q1 and Q2 have on-state re-
sistances Ron1 and Ron2, respectively. The transformer
T1 has secondary side-related leakage inductance Ls,
magnetizing inductance Lm, primary winding resis-
tance Rpri, secondary winding resistance Rs, and turns
ratio 1 : n. The diode D1 is modeled with on-state re-
sistance RD and forward bias voltage VD. Components
listed above are internal parts of the proposed model.
The input voltage vin(t), output voltages vout1(t) and
vout2(t), and the corresponding load networks (R1/C1
and R2/C2) are connected externally to the model.
The converter has switching frequency Fsw = 1/T .

Figure 1: Fly-buck converter with two outputs.

The main waveforms are shown in Fig. 2.
The switching period is divided into three parts, and
the first interval d1 is the time when leakage induc-
tance Ls resets. The switch Q1 is on, and the Q2 is
off. The diode D1 is forward-biased. The second in-
terval d2 is the time when the diode D1 blocks, Q1 is
on, and Q2 is off. The third interval d3 is the time

when Q1 is off, Q2 is on, and D1 conducts. The duty
cycle is determined as d = d1 +d2 and 1−d = d3.

Figure 2: Waveforms of the fly-buck converter.

States of the converter at each time interval d1−d3
are presented in Fig. 3. Using the small ripple approx-
imation average, voltage across the inductor can be
obtained:

Lm
d〈iL(t)〉T

dt
= 〈vin(t)〉T d−〈vout1(t)〉T

−〈iL(t)〉T (Ron1 d +Ron2 (1−d)+Rpri)+

〈iout2(d1)
(t)〉

T
(Ron1 +Rpri)n+

〈iout2(d3)
(t)〉

T
(Ron2 +Rpri)n, (1)

where 〈x(t)〉T represents the average value of x over
the switching period T . The currents iout2(d1)

(t) and
iout2(d3)

(t) are artificially shown in Fig. 2 separately, so
iout2(t)= iout2(d1)

(t)+iout2(d3)
(t) due to iout2(d2)

(t)=0.
The average values of these currents will be obtained
later.

By using the charge balance approach, the average
currents for C1 and C2 can be found:

Figure 3: Equivalent circuits of the converter for different
time intervals.

C1
d 〈vout1(t)〉T

dt
=−〈iout2〉T n−〈vout1(t)〉T/R1+〈iL〉T

(2)

C2
d 〈vout2(t)〉T

dt
=−〈vout2(t)〉T/R2 + 〈iout2(t)〉T

(3)

The input voltage source’s average current can be ob-
tained as follows:

〈iin(t)〉T = 〈iL(t)〉T d−〈iout2d1(t)〉T n (4)

To build the final model, the average currents
〈iout2d1(t)〉T and 〈iout2d3(t)〉T must be obtained. It can
be seen from Fig. 2 that current iout2d3(t) is an ex-
ponential process of magnetizing leakage inductance
Ls. Fig. 3c can be used to find an analytical solu-
tion for the 〈iout2d3(t)〉T average current. The transient
process during one switching period is considered.
Variables iL(t), vout1(t) and vout2(t) can be replaced
with constant sources for one switching period due to
the small ripple approximation. The initial current in

the Ls inductor is zero, so a solution for the peak and
average currents can be found:

max(iout2)T =
Ed3(t)
Rd3 n

(
1−e−

Rd3 n2 to f f
Ls

)
(5)

〈iout2d3(t)〉T =
Ed3(t)Fsw

Rd3 n

(
Ls
(
1−e−

Rd3 n2 to f f
Ls

)
Rd3 n2 +to f f

)
,

(6)

where Rd3 = Ron2+Rpri+(Rs +RD)/n2, to f f = (1−
d)/Fsw and

Ed3(t) = 〈vout1(t)〉+〈iL(t)〉(Ron2+Rpri)−
(〈vout2(t)〉+VD)

n
.

A similar approach can be used to find
〈iout2d1(t)〉T ’s average current during the d1 interval.
Fig. 3a represents an equivalent circuit for this inter-
val. Transient process of the leakage inductance reset
is also considered in one particular switching period.
The initial current in the Ls inductor is max(iout2)T ,
found from (5), and then it resets to zero current.
Thus, the solution is obtained as follows:

〈iout2d1(t)〉T =
Ed3(t)Fsw Ls

n3 Rd1 Rd3

(
1− e−

n2 to f f Rd3
Ls

)
+

Ed1(t)Fsw Ls

n3 Rd1
2 ln

(
1− Ed3(t)Rd1

Ed1(t)Rd3

(
1− e−

n2 to f f Rd3
Ls

))
,

(7)

where Ed1(t)=Ed3(t)−〈vin(t)〉−〈iL(t)〉(Ron2−Ron1),
Rd1 = Ron1 +Rpri +(Rs +RD)/n2. Using (1)–(3),
the schematic of the fly-buck converter model can be
constructed as shown in Fig. 4. The 〈iout2d1(t)〉T and
〈iout2d3(t)〉T currents ((6) and (7)) and realization of
(3) are implemented by the Gd1 and Gd2 arbitrarily
behavior current sources. The E3 source, along with
L1, realizes (1). The G3 source is responsible for (4),
while the G4 source implements (2). The ideal diodes
D1 and D2 improve the convergence of the model by
blocking negative voltages on the second output.

The model has next pins to connect to external cir-
cuits: node ‘Vin’ — input voltage, node ‘Vout1’ —
first output (non-isolated), node ‘Vout2’ — second
output (isolated) and node ‘d’ — duty cycle control
input (0.0...1.0 range). The reference zero potential
for the primary side is connected to the global ‘0’
net, and the secondary side’s ground potential is con-
nected using a GND SEC pin.

The sub-model netlist can be found in
Fig. 5 and (Zaikin, 2019).

Figure 4: Fly-buck converter average model schematic.

.SUBCKT FLY_BUCK_AVG Vin Vout1 Vout2 GND_SEC d
+PARAMS: Ron1=12m Ron2=12m VD=1.8 RD=0.2 Rs=0.07
+Ls=7.5u Rpri=10m Lm=3.8u n=5 Fsw=100k
.param Rd1={Ron1+Rpri+(Rs+RD)/n**2}
.param Rd3={Ron2+Rpri+(Rs+RD)/n**2}
L1 2 Vout1 {Lm}
Gd1 0 5 VALUE={(V(Ed1)*Fsw*Ls*log(1-(V(Ed3)*Rd1*
+(exp(-(Rd3*n**2*V(toff))/Ls)-1))/(V(Ed1)*Rd3)))/(Rd1**2*
+n**3)+(V(Ed3)*Fsw*Ls*(exp(-(Rd3*n**2*V(toff))/Ls)-1))/
+(Rd1*Rd3*n**3)}
Gd3 0 4 VALUE={(V(Ed3)*Fsw*(v(toff)+(Ls*(exp(-
+(Rd3*n**2*v(toff))/Ls)-1))/(Rd3*n**2)))/(Rd3*n)}
G3 0 Vin VALUE={I(Vd1)*n}
Ed1 Ed1 0 VALUE={V(Ed3)-V(Vin)-I(L1)*(Ron2-Ron1)}
Ed3 Ed3 0 VALUE={V(Vout1)+I(L1)*(Ron2+Rpri)-
+(V(Vout2,0)+VD)/n}
E3 2 1 VALUE={-(V(d)*Ron1+(1-V(d))*Ron2+Rpri)*I(L1)+
+n*(Ron1+Rpri)*I(Vd1)+n*(Ron2+Rpri)*I(Vd3)}
G4 Vout1 0 VALUE={(I(Vd1)+I(Vd3))*n}
D1 0 3 Dbreak
Etoff toff 0 VALUE={(1-V(d))/Fsw}
D2 3 Vout2 Dbreak
Vd3 4 3 0
Vd1 5 3 0
EdcXFMR 6 0 VALUE={V(Vin)*V(d)}
GdcXFMR Vin 0 VALUE={I(VdcXFMR)*V(d)}
RpdcXFRM Vin 0 1MEG
RsdcXFRM 1 7 1u
VdcXFMR 6 7 0
.ends FLY_BUCK_AVG
.model Dbreak D Is=1e-14 Cjo=.1pF Rs=1m N=0.01

Figure 5: The fly-buck converter LTSpice/PSpice sub-
model netlist. Copy-paste is possible from pdf version of
this paper.

Figure 6: Fly-buck converter simulation setup.

3 RESULTS

The proposed model was simulated and com-
pared to switching modeling, along with proto-

type measurement. The parameters for simula-
tion and testing were VD = 1.8 V, RD = 0.2 Ohm
(C3D06060A two in series), Ron1 =Ron2=12 mOhm
(IPB117N20NFD), Fsw=100 kHz, Ls=7.5 uH,
Rs=0.07 Ohm, Lm=3.8 uH, Rpri=10 mOhm, and
n=5. The external circuit contained the input cable’s
40 uH inductance, and capacitance at the input was
7.92 mF (25 mOhm ESR), R1=100 kOhm, C1=940 uF
(35 mOhm ESR), R2=75 Ohm, C2=2.35 uF and the in-
put voltage vin=50 V. The circuit for simulation and
measurement is shown in Fig. 6. The simulation re-
sults are presented in Fig. 7 and Fig. 8.

Figure 7: Simulation results. The vout2 step response on
the duty cycle changed from 0.4 to 0.5.

Figure 8: Simulation results. The transfer function v̂out2( f )

of the output voltage compared to a control.

The setup for testing is shown in Fig. 9. The mea-
surement results are presented in Fig. 10.

Figure 9: Fly-buck converter test setup.

Figure 10: Measurement results. The vout2 step response on
the duty cycle changed from 0.4 to 0.5.

4 CONCLUSION

The proposed model can be simulated for large and
small signal modeling in time or frequency domains.
The model accounts for parasitics of semiconductors
and magnetics so losses and precise behavior can be
predicted. A listing of the SPICE model was pre-
sented and it can be used for the static and dynamic
behavior analysis of the fly-buck converter.

